CONFIG(debug, debug|release) {
    PREFIXNAME = handipaint-debug
} else {
    PREFIXNAME = handipaint-release
}

win32 {
    system(echo win32 prefix)
    PREFIX = C://$$PREFIXNAME
    MINGW_LIB_DIR = C:/QtSDK/Desktop/Qt/4.8.0/mingw/bin
    QT_SRC_3RDPARTY = C:/QtSDK/QtSources/4.8.0/src/3rdparty
}

unix {
    system(echo unix prefix)
    PREFIX = /tmp/$$PREFIXNAME
    EXTRA_LIB_DIR = /needtofindrightpath
    QT_SRC_3RDPARTY = /needtofindright3rdpartypathonunix
}

QT       += core gui webkit network

TARGET = HandiPaint2012
TEMPLATE = app

contains(QT_CONFIG, zlib):include($$QT_SRC_3RDPARTY/zlib.pri)
else:include($$QT_SRC_3RDPARTY/zlib_dependency.pri)

SOURCES += src/main.cpp\
        src/HPMainWindow.cpp \
    src/HPGraphicsScene.cpp \
    src/HPTools.cpp

HEADERS  += src/HPMainWindow.h \
    src/HPGraphicsScene.h \
    src/HPTools.h

FORMS    += src/HPMainWindow.ui

RESOURCES += \
    HPResources.qrc

INCLUDEPATH += $$PREFIX/include

LIBS += -L$$PREFIX
LIBS += -lquazip

OTHER_FILES += \
    HandiPaintConfig.ini

target.path = $$PREFIX

config.files = HandiPaintConfig.ini
config.path = $$PREFIX

win32 {
    system(echo win32)
    extralibs.files = $$MINGW_LIB_DIR/mingwm10.dll
    extralibs.files += $$MINGW_LIB_DIR/libgcc_s_dw2-1.dll
CONFIG(debug, debug|release) {
    system(echo debug dll)
    extralibs.files += $$MINGW_LIB_DIR/QtCored4.dll
    extralibs.files += $$MINGW_LIB_DIR/QtGuid4.dll
    extralibs.files += $$MINGW_LIB_DIR/QtNetworkd4.dll
    extralibs.files += $$MINGW_LIB_DIR/QtWebKitd4.dll
    extralibs.files += $$MINGW_LIB_DIR/phonond4.dll
} else {
    system(echo release dll)
    extralibs.files += $$MINGW_LIB_DIR/QtCore4.dll
    extralibs.files += $$MINGW_LIB_DIR/QtGui4.dll
    extralibs.files += $$MINGW_LIB_DIR/QtNetwork4.dll
    extralibs.files += $$MINGW_LIB_DIR/QtWebKit4.dll
    extralibs.files += $$MINGW_LIB_DIR/phonon4.dll
}
    extralibs.path = $$PREFIX
}

INSTALLS += target \
    config \
    extralibs
